Rozwiązywanie układu równań liniowych z macierzą symetryczną dodatnio określoną metodą Choleskiego

Adam Szczepański

1. Zastosowanie

Niech $[A] \in \mathbb{IR}^{n \times n}$ będzie macierzą przedziałów spełniającą $[A] = [A]^T$. Niech $[b] \in \mathbb{IR}$ będzie dane. Chcemy ograniczyć zestaw rozwiązań

$$S = \{x | Ax = b, A \in [A], A = A^T, b \in [b]\}$$

przez wektor przedziałów $[x]$.

2. Opis metody

Metoda polega na znalezieniu macierzy dolnotrójkątnej

$$\begin{bmatrix} [l_{11}] & 0 & \dots & 0 \\ [l_{21}] & [l_{22}] & \dots & 0 \\ \dots & \dots & \dots & \dots \\ [l_{n1}] & [l_{n2}] & \dots & [l_{nn}] \end{bmatrix}$$

takiej, że $L = AA^T$, a następnie obliczeniu rozwiązania na podstawie wzorów

$$[y_i] = \frac{[b_i] - \sum_{j=1}^{i-1} [l_{ij}][y_j]}{[l_{ii}]}, i = 1,2, \dots, n$$

$$[x_i] = \frac{[y_i] - \sum_{j=i+1}^{n} [l_{ji}][x_j]}{[l_{ii}]}, i = n, n-1, \dots, 1$$

3. Wywołanie procedury

symposmatrix_ia(n, a, b, x, st)

4. Dane

n – liczba równań układu
a – tablica z wartościami elementów macierzy $[A]$ układu ($a[i,j]$ zawiera przedział $[a_{ij}], i, j = 1,2, \dots, n$); macierz powinna być symetryczna i dodatnio określona
b – tablica z wartościami elementów macierzy $[b]$ układu ($b[i]$ zawiera przedział $[b_i], i = 1,2, \dots, n$)

5. Wynik

x – tablica z rozwiązaniem układu ($x[i]$ zawiera przedział $[x_i], i = 1,2, \dots, n$)

6. Inne parametry

st – zmienna wyjściowa, której przypisywany jest status obliczeń, tzn:
1, jeżeli *n < 1*
2, jeżeli macierz układu nie jest symetryczna
3, jeżeli macierz układu nie jest dodatnio określona
0, w przeciwnym wypadku

7. Typy parametrów

Integer: n, st
matrix_ia: a
vector_ia: b, x

8. Identyfikatory nielokalne

vector_ia – nazwa typu tablicowego $[q_1 \dots q_n]$ o elementach typu *interval*
matrix_ia – nazwa typu tablicowego $[q_1 \dots q_n, q_1 \dots q_n]$ o elemtach typu *interval*

9. Treść procedury

```
procedure symposmatrix_ia (n      : Integer;
                        var a  : matrix_ia;
                        b      : vector_ia;
                        var x  : vector_ia;
                        var st : Integer);
var i,j,k : Integer;
    z,tmp,a1 : interval;
begin
  a1:=int_read('1');
  st:=0;
  if n<1
    then st:=1;
  for i:=2 to n do
    for j:=1 to i-1 do
      if ((a[i,j].a<>a[j,i].a) or (a[i,j].b<>a[j,i].b))
        then st:=2;
  if st=0
    then begin
           i:=0;
           repeat
             i:=i+1;
             for j:=i to n do
               begin
```

```
                 z:=a[i,j];
                 for k:=i-1 downto 1 do
                   z:=(isub(z,imul(a[j,k],a[i,k])));
                 if i=j
                   then if ((z.a>0) and (z.b>0))
                          then begin
                            tmp.a:=sqrt(z.a);
                            tmp.b:=sqrt(z.b);
                            a[i,i]:=idiv(a1,tmp)
                          end
                          else st:=3
                   else a[j,i]:=imul(z,a[i,i])
               end
           until (i=n) or (st=3);
           if st=0
             then begin
                    for i:=1 to n do
                      begin
                        z:=b[i];
                        for j:=i-1 downto 1 do
                          z:=isub(z,imul(a[i,j],x[j]));
                        x[i]:=imul(z,a[i,i])
                      end;
                    for i:=n downto 1 do
                      begin
                        z:=x[i];
                        for j:=i+1 to n do
                          z:=isub(z,imul(a[j,i],x[j]));
                        x[i]:=imul(z,a[i,i])
                      end
                  end
         end
end;
```

10. Przykłady

a. Dane:

$$n = 1$$
$$[A] = ([1,2]), [b] = ([2,2])$$

Wyniki:

$x = ([9.9999999999999999E-0001, 2.0000000000000000E+0000])$

$st = 0$

b. Dane:

$$n = 2$$

$$[A] = \begin{pmatrix} 4 & [-1,1] \\ [-1,1] & 4 \end{pmatrix}, [b] = \begin{pmatrix} 6 \\ 6 \end{pmatrix}$$

Wyniki:

$$x = \begin{pmatrix} [9.9999999999999999\mathrm{E}-0001, 2.0000000000000001\mathrm{E}+0000] \\ [1.0588235294117647\mathrm{E}+0000, 2.0000000000000001\mathrm{E}+0000] \end{pmatrix}$$

$st = 0$

c. Dane:

$$n = 2$$

$$[A] = \begin{pmatrix} [1,4] & [0,1] \\ [0,1] & 3 \end{pmatrix}, [b] = \begin{pmatrix} 2 \\ [0,2] \end{pmatrix}$$

Wyniki:

$$[x] = \begin{pmatrix} [-1.0842021724855045\mathrm{E}-0019, 3.0000000000000001\mathrm{E}+0000] \\ [-1.0000000000000001\mathrm{E}+0000, 1.0000000000000001\mathrm{E}+0000] \end{pmatrix}$$

$st = 0$

d. Dane:

$$n = 3$$

$$[A] = \begin{pmatrix} 2 & 1 & 2 \\ 1 & 2 & -1 \\ 2 & -1 & 2 \end{pmatrix}, [b] = \begin{pmatrix} 1 \\ 2 \\ 0 \end{pmatrix}$$

Wyniki:

$st = 3$